

Programmation dynamique

La programmation dynamique s'applique le plus souvent aux problèmes d'optimisation pour lesquels on doit faire un ensemble de choix pour arriver à une solution optimale. À mesure que de nouvelles options sont choisies, des sous-problèmes de la même forme apparaissent souvent. La programmation dynamique est efficace lorsqu'un sous-problème donné peut apparaître pour plusieurs suites d'options différentes. Le principe est de stocker la solution à chaque sous-problème au cas où il réapparaîtrait. Dans ce sujet, nous étudions deux tels exemples.

1 Plus longue sous-suite commune

On considère une suite finie $x = (x_0, \dots, x_{m-1})$ formée de m éléments d'un ensemble E . Une *sous-suite* de x de longueur k est une suite x' obtenue à partir de x en supprimant $m - k$ éléments tout en conservant l'ordre : $x' = (x_{\sigma(0)}, \dots, x_{\sigma(k-1)})$ avec $0 \leq \sigma(0) < \dots < \sigma(k-1) < m$. On dit que z est une sous-suite commune de x et y si z est une sous-suite de x et une sous-suite de y .

Nous représenterons en Caml les suites finies par des tableaux. Nous cherchons maintenant à trouver une plus longue sous-suite commune à deux suites $x = (x_0, \dots, x_{m-1})$ et $y = (y_0, \dots, y_{n-1})$. Une méthode naïve consisterait à construire l'ensemble des sous-listes de x et de y , puis à calculer leur intersection.

► **Question 1** Évaluez la complexité, en fonction de la taille des listes x et y de cette méthode.

Dans un premier temps, nous nous limitons au problème du calcul de la *longueur* d'une plus longue sous-suite commune de deux suites x et y . On note $\ell(i, j)$ la longueur de la plus longue sous suite commune des suites (x_0, \dots, x_{i-1}) et (y_0, \dots, y_{j-1}) . La longueur recherchée est donnée par $\ell(m, n)$ (où m et n sont les longueurs respectives de x et y). De plus, ℓ vérifie la relation de récurrence suivante :

$$\ell(i, j) = \max(\ell(i-1, j-1) + \delta(x_{i-1}, y_{j-1}), \ell(i, j-1), \ell(i-1, j))$$

où $\delta(x_i, y_j) = 1$ si $x_i = y_j$ et 0 sinon.

Il est facile de vérifier que si on utilise cette relation pour écrire une fonction récursive calculant la valeur de $\ell(i, j)$, on obtiendrait à nouveau une fonction dont le coût serait exponentiel en la taille de l'entrée : l'algorithme récursif rencontre en effet chaque sous-problème de nombreuses fois dans différentes branches de l'arbre récursif, sans réutiliser toutefois les calculs déjà effectués. La programmation dynamique contourne ce problème en stockant

les résultats intermédiaires dans un tableau à deux dimensions auxiliaire : la case (i, j) de cette matrice contenant la longueur d'une plus longue sous-liste commune de (x_0, \dots, x_{i-1}) et (y_0, \dots, y_{j-1}) , c'est-à-dire $\ell(i, j)$.

► **Question 2** Expliquez comment on peut remplir de proche le tableau auxiliaire.

► **Question 3** Déduisez-en une fonction qui permette de calculer la longueur de la plus longue sous-suite commune à deux suites. Votre fonction aura une complexité dominée par le produit des longueurs des deux listes passées en argument.

```
val subseq_length: 'a vect -> 'a vect -> int
```

On souhaite maintenant obtenir une des sous-suites communes de longueur maximale de deux suites. Pour cela, remarquons que la valeur de chaque case du tableau auxiliaire est obtenue à partir de celle de l'une de ses trois voisines : celle située au dessus, celle située à gauche ou celle située en diagonale en haut à gauche. En conservant (dans un second tableau) pour chaque case une marque indiquant à partir de quelle voisine sa valeur a été calculée, il est possible à la fin de l'algorithme de reconstituer une plus longue sous-suite commune en parcourant le tableau obtenu en suivant les directions indiquées par les marques comme illustré sur la figure suivante :

		j	0	1	2	3
i		y_{j-1}	a	b	c	
0	x_{i-1}	0	0	0	0	
1	a	0	↖ 1	← 1	← 1	
2	a	0	↖ 1	↑ 1	↑ 1	
3	c	0	↑ 1	↑ 1	↖ 2	
4	b	0	↑ 1	↖ 2	↑ 2	

Pour représenter les marques, vous pouvez définir en Caml le type suivant :

```
type mark =
  | North
  | West
  | NorthWest
;;
```

► **Question 4** Écrivez une fonction `subseq` qui retourne une des plus longues sous-suites commune de deux suites.

```
val subseq: 'a vect -> 'a vect -> 'a vect
```

2 Multiplication d'une suite de matrices

Notre deuxième exemple de programmation dynamique est un algorithme qui résoud le problème de la multiplication d'une suite de matrices. On considère une suite de n matrices d'entiers (M_0, M_1, \dots, M_{n-1}) à multiplier, i.e. on souhaite calculer le produit matriciel $M_0 M_1 \dots M_{n-1}$. Il est possible d'évaluer ce produit en utilisant l'algorithme classique permettant de multiplier deux matrices, après avoir placé des parenthèses pour préciser l'ordre dans lequel les produits binaires doivent être effectués. La multiplication des matrices étant associative, tous les parenthésages aboutissent à une même valeur du produit. Cependant, la manière dont une suite de matrices est parenthésée peut avoir un impact crucial sur le coût d'évaluation du produit.

► **Question 5** Écrivez une fonction qui calcule le produit de deux matrices d'entiers. (Dans le cas où la taille des matrices n'est pas compatible, vous lèverez une exception.) Précisez le coût (en nombre de multiplication d'entiers) du produit de deux matrices en fonction de leur taille.

```
val prod: int vect vect -> int vect vect -> int vect vect
```

► **Question 6** Donnez un exemple de dimensions pour trois matrices M_0, M_1 et M_2 tel que le coût du calcul de $(M_0 M_1) M_2$ soit différent de celui de $M_0 (M_1 M_2)$. Montrez que le rapport entre ces deux coûts peut être arbitrairement grand.

Pour le problème qui nous intéresse, on peut représenter la suite de matrices (M_0, \dots, M_{n-1}) par un tableau de $n+1$ entiers d tels que, pour tout i , la matrice M_i a pour dimensions $(d(i), d(i+1))$. On note également, pour $i \leq j$, $M_{i,j}$ le produit $M_i M_{i+1} \dots M_j$ et $p_{i,j}$ le nombre minimum de multiplications d'entiers nécessaires pour le calcul de ce produit.

► **Question 7** Montrez que si $i < j$ alors $p_{i,j}$ est égal au minimum de l'ensemble $\{p_{i,k} + p_{k+1,j} + d(i) \times d(k+1) \times d(j+1) \mid i \leq k < j\}$.

► **Question 8** Écrivez une fonction `min_fun` prenant pour arguments une fonction f de type $\text{int} \rightarrow \text{int}$ et deux entiers i_1 et i_2 (tels que $i_1 \leq i_2$). Votre fonction calculera l'entier i compris entre i_1 et i_2 tel que $f(i)$ soit minimal.

```
val min_fun: (int -> int) -> int -> int -> int
```

► **Question 9** Déduisez-en une fonction qui calcule le nombre minimal de multiplications entières à effectuer pour calculer le produit d'une suite de matrices. (Votre fonction prendra pour argument le tableau des dimensions d .)

```
val optprod: int vect -> int
```

La fonction que vous venez d'écrire a probablement un temps d'exécution exponentiel, ce n'est pas meilleur que la méthode directe consistant à tester chaque manière de parenthéser le produit : l'algorithme récursif rencontre en effet chaque sous-problème de nombreuses fois dans différentes branches de l'arbre récursifs, sans réutiliser toutefois les calculs déjà effectués.

L'approche par *programmation dynamique* consiste à calculer incrémentalement les valeurs de $p_{i,j}$ dans l'ordre des $j - i$ croissants et à stocker les valeurs obtenues dans un tableau à deux dimensions. Ainsi, le contenu de chaque case du tableau n'est évalué qu'une seule fois et son calcul ne fait appel qu'à des cases déjà calculées. Voici le tableau obtenu pour une suite de quatre matrices de dimensions $2 \times 4, 4 \times 3, 3 \times 1$ et 1×5 :

$p_{i,j}$	0	1	2	3
0	0	24	20 <small>(k=1)</small>	30 <small>(k=2)</small>
1		0	12 <small>(k=2)</small>	32 <small>(k=2)</small>
2			0	15
3				0

► **Question 10** Écrivez une fonction `optprod_dyn` qui calcule le nombre minimal de multiplications entières à effectuer pour calculer le produit d'une liste de matrices en utilisant cette méthode.

```
val optprod_dyn: int vect -> int
```

► **Question 11** Évaluez la complexité de la fonction `optprod_dyn`.

On souhaite maintenant obtenir la forme du parenthésage qui permet d'atteindre le minimum. Il suffit pour cela d'utiliser un tableau à deux dimensions auxiliaire s tel que $s_{i,j}$ contienne la valeur de k pour laquelle le minimum défini à la question 7 a été atteint. Pour représenter un tel parenthésage, on peut utiliser un arbre binaire dont les feuilles sont étiquetées par les numéros des matrices :

```
type tree =
  | Matrix of int
  | Product of tree * tree
;;
```

► **Question 12** Déduisez-en une fonction qui construit l'arbre représentant le meilleur parenthésage pour calculer le produit d'une liste de matrices.

```
val optprod_tree: int vect -> tree
```

Programmation dynamique

Un corrigé

► **Question 2** On peut remplir le tableau dans l'ordre des $i + j$ croissants : le calcul de $\ell(i, j)$ fait en effet appel aux valeurs de $\ell(i - 1, j)$, $\ell(i, j - 1)$ et $\ell(i - 1, j - 1)$.

► **Question 3**

```
let subseq_length x y =
  let m = vect_length x
  and n = vect_length y in
  let s = make_matrix (m+1) (n+1) 0 in

  for i = 1 to m do
    for j = 1 to n do
      s.(i).(j) <- max
        (s.(i-1).(j-1) + (if x.(i-1) = y.(j-1) then 1 else 0))
        (max s.(i-1).(j) s.(i).(j-1))
    done;
  done;
  s.(m).(n)
;;
```

► **Question 4** Notre fonction comporte deux parties. La première remplit incrémentalement les matrices s et t . Elle est analogue à la fonction `subseq_length`. La deuxième partie parcourt la matrice t de manière à retrouver construire une des plus longues sous-séquences communes.

```
let subseq x y =
  let m = vect_length x
  and n = vect_length y in
  let s = make_matrix (m+1) (n+1) 0
  and t = make_matrix (m+1) (n+1) NorthWest in

  for i = 1 to m do
    for j = 1 to n do
      if x.(i-1) = y.(j-1) then s.(i).(j) <- s.(i-1).(j-1) + 1
      else begin
        if s.(i-1).(j) > s.(i).(j-1)
        then (s.(i).(j) <- s.(i-1).(j); t.(i).(j) <- North)
        else (s.(i).(j) <- s.(i).(j-1); t.(i).(j) <- West)
      end
    done;
  done;

  let r = make_vect s.(m).(n) x.(0) in
  let i = ref m and j = ref n and k = ref s.(m).(n) in

  while !k > 0 do
    match t.(!i).(!j) with
    | North -> i := !i - 1
    | West -> j := !j - 1
    | NorthWest ->
      i := !i - 1;
      j := !j - 1;
      k := !k - 1;
      r.(!k) <- x.(!i)
  done;
  r
;;
```

► **Question 5** La fonction `dim` détermine les dimensions d'une matrices.

```
let dim m =
  let n = vect_length m in
  if n = 0 then (0, 0) else (n, vect_length m.(0))
;;
```

La fonction `prod` calcule le produit de deux matrices entières.

```
let prod m1 m2 =
  let n1, p1 = dim m1
  and n2, p2 = dim m2 in

  if p1 <> n2 then invalid_arg "prod";

  let m = make_matrix n1 p2 0 in

  for i = 0 to n1 - 1 do
    for j = 0 to p2 - 1 do
      for k = 0 to p1 - 1 do
        m.(i).(j) <- m.(i).(j) + m1.(i).(k) * m2.(k).(j)
      done
    done
  done;

  m
;;
```

Notons (n_1, p_1) et (n_2, p_2) les dimensions respectives des matrices M_1 et M_2 . Le produit $M_1 M_2$ est défini si et seulement si $p_1 = n_2$. La fonction `product` ci-dessus le calcule alors en effectuant $n_1 p_1 p_2$ produits d'entiers.

► **Question 6** Considérons une suite de trois matrices de dimensions suivantes :

$$M_1 : 2 \times n$$

$$M_2 : n \times 2$$

$$M_3 : 2 \times n$$

Si on effectue le produit en utilisant le parenthésage $(M_1 M_2) M_3$, on effectue $2 \times n \times 2$ multiplications puis $2 \times 2 \times n$, soit au total $8n$ multiplications d'entiers. Au contraire, si on considère le parenthésage $M_1 (M_2 M_3)$ alors $4n^2$ multiplications sont nécessaires. Le rapport entre les coûts deux alternatives est $n/2$.

► Question 8

```
let rec min_fun f i1 i2 =
  if i1 = i2 then
    i1
  else
    let i = min_fun f (i1 + 1) i2 in
    if f i < f i1 then i else i1
;;
```

► **Question 9** Notre fonction utilise une sous-fonction `p` qui permet de calculer récursivement la valeur de $p_{i,j}$ pour $i \leq j$.

```
let optprod d =
  let n = vect.length d - 1 in
  let rec p i j =
    let f k = p i k + p (k+1) j + d.(i) * d.(k+1) * d.(j+1) in
    if i = j then 0 else f (min_fun f i (j-1))
  in
  p 0 (n - 1)
;;
```

► Question 10

```
let optprod_dyn d =
  let n = vect.length d - 1 in
  let p = make_matrix n n 0 in
  for l = 1 to n - 1 do
    for i = 0 to n - l - 1 do
      let j = l + i in
      let f k =
        p.(i).(k) + p.(k+1).(j) + d.(i) * d.(k+1) * d.(j+1)
      in
      let k = min_fun f i (j-1) in
      p.(i).(j) <- f k
    done
  done;
  p.(0).(n-1)
;;
```

► **Question 12** Notre fonction `optprod_tree` se décompose en deux parties. La première est analogue au corps de la fonction `optprod`, à ceci près que, à chaque itération, elle stocke la valeur de k pour laquelle le minimum est trouvé dans un tableau auxiliaire s . La seconde partie est formée d'une fonction récursive auxiliaire qui construit l'arbre de parenthésage à partir de ce tableau.

```
let optprod_tree d =
  let n = vect.length d - 1 in
  let p = make_matrix n n 0 in
  let s = make_matrix n n (-1) in
  for l = 1 to n - 1 do
    for i = 0 to n - l - 1 do
      let j = l + i in
      let f k = p.(i).(k) + p.(k+1).(j) + d.(i) * d.(k+1) * d.(j+1) in
      let k = min_fun f i (j-1) in
      p.(i).(j) <- f k;
      s.(i).(j) <- k
    done
  done;
  let rec aux i j =
    if i = j then (Matrix i)
    else Product (aux i s.(i).(j), aux (s.(i).(j) + 1) j)
  in
  aux 0 (n-1)
;;
```